

SUB - SEARCH

Contents

Instructions	1
Flowchart	2
Sample Run	3
Program Listing	4-5
Notes	6
10x10x10 Grid Area	7

INSTRUCTIONS

IN THE 'SUB SEARCH' PROGRAM YOU ARE CAPTAIN OF THE DESTROYER USS BINARY. YOU ARE GIVEN ORDERS TO SEEK AND DESTROY A STATIONARY (IT DOESN'T MOVE) ENEMY SUB , YOU SEARCH FOR IT IN A 3-DIMENSIONAL CUBE OF WATER. THE SIZE OF THE CUBE IS SELECTED BY YOU , AND SO IS THE NUMBER OF SHOTS. IF YOU ARE CONFUSED , TAKE A LOOK AT THE SAMPLE RUN ON PAGE #3. AFTER YOU HAVE ENTERED AREA DIMENSIONS (CUBE SIZE) THE PROGRAM WILL ASK YOU 'DO YOU WANT INSTRUCTIONS?' , BEFORE ANSWERING THIS EARTSHAKING QUESTION , PLEASE NOTE THAT THE INSTRUCTIONS BUILT INTO THE PROGRAM ARE NOT ANYWHERE NEAR AS DETAILED AS THESE. BUT FOR THE FIRST TIME TYPE A 'Y'. OH , YES , I FORGOT TO TELL YOU TO ANSWER QUESTIONS WITH A 'Y' FOR YES AND A 'N' FOR NO.

NOW THAT YOU HAVE STARTED READING THE INSTRUCTIONS YOU WILL SEE IN THE SECOND SCREENFUL THAT THE INSTRUCTIONS MENTION 'X,Y,Z' COORDINATES', THESE ARE NOTHING BUT 3-DIMENSIONAL COORDINATES. X,Y,Z COORDINATES ARE NOTHING BUT THE REGULAR X,Y (2-DIMENSIONAL) COORDINATES WITH THE NEXT DIMENSION (THIRD DIMENSION/Z COORDINATE) ADDED. NOW THAT I'VE EXPLAINED ALL THAT, YOU CAN GO ON TO THE NEXT PARAGRAPH ON PLAYING THE GAME.

WHEN THE PROGRAM ASKS YOU 'EAST/WEST:' IT IS REQUESTING THAT YOU TYPE A NUMBER (FROM 0 TO THE X DIMENSION -1, IF YOU ARE CONFUSED CONSULT THE SAMPLE RUN.) TO REPRESENT YOUR GUESS FOR THE X COORDINATE OF THE SUB. NEXT THE PROGRAM WILL ASK YOU 'NORTH/SOUTH:', RESPOND WITH YOUR GUESS FOR THE Y COORDINATE OF THE SUB. TO END YOUR DATA INPUTTING FOR THIS TRIAL, THE PROGRAM WILL ASK YOU 'DEPTH:', NOW TYPE IN YOUR GUESS FOR THE Z COORDINATE OF THE SUB. AT THIS POINT, THE PROGRAM WILL CLEAR THE SCREEN AND PRINT OUT THE LOCATION OF THE EXPLOSION COMPARED TO THE LOCATION OF THE SUB. (SEE SAMPLE RUN FOR DETAILS). IT WILL ALSO PRINT OUT THE LOCATION OF THE LAST COORDINATES (THE ONES INPUTTED DURING THE LAST TRIAL). NEXT IT WILL GO THROUGH THE X,Y,Z COORDINATE INPUT ROUTINE (UNLESS THE LAST COORDINATES WERE THE EXACT SAME LOCATION AS THE SUB), AND THEN GO THROUGH THE EXPLOSION POINT RELATIVE TO SUB COORDINATES ROUTINE AND THEN THE LAST COORDINATES PRINT OUT ROUTINE. THIS WILL GO ON TILL EITHER ONE OF TWO THINGS HAPPEN ;


1. YOU WILL FINALLY GET THE SUB, AND THE PROGRAM WILL CONGRATULATE YOU.
2. YOU WILL RUN OUT OF SHOTS AND GET TORPEDOED BY THE SUB.

ON PAGE #7 THERE IS A DRAWING OF A 10 x 10³-DIMENSIONAL GRID AREA.

xio

[illegible]

FLOWCHART



INIT , LINES 1000-1080

AREA DIMS , LINES 1090-1160

INSTRUCTIONS? , LINES 1210-1230

PRINT OUT INSTRUCTIONS , LINES 1240-1430

A,B,C = SUB LOCATION , LINES 1450-1470

START TRIAL LOOP , LINES 1480-1480

X,Y,Z = PLAYERS GUESS , LINES 1510-1530

X,Y,Z = A,B,C , LINES 1540-1540

START NEXT TRIAL , LINES 1560-1590

TRIALS ALL USED , LINES 1600-1630

IT'S A HIT , LINES 1640-1670

ANOTHER GAME? , LINES 1680-1730

SONAR ROUTINE , LINES 1740-1840

RESET SIMULATE , LINES 1850-1910

Note:

IF X, Y, Z = A, B, C

GOTO 1640

SAMPLE RUN

A LONG LINE REPRESENTS THE CLEARING OF THE SCREEN AND PRINTING THE NEXT SCREENFUL OF INFORMATION.

RUN

AREA DIMENSIONS:
EAST/WEST: 10
NORTH/SOUTH: 10
DEPTH: 10
HOW MANY SHOTS? 5

DO YOU WANT INSTRUCTIONS? (NO)

TRIAL #1
EAST/WEST: 6
NORTH/SOUTH: 5
DEPTH: 5

SONAR SENSES EXPLOSION POINT WAS NORTH AND TOO LOW.
LAST COORDINATES = 6,5,5

TRIAL #2
EAST/WEST: 6
NORTH/SOUTH: 3
DEPTH: 3

SONAR SENSES EXPLOSION POINT WAS NORTH AND TOO LOW.
LAST COORDINATES = 6,3,3

TRIAL #3
EAST/WEST: 6
NORTH/SOUTH: 1
DEPTH: 1

SONAR SENSES EXPLOSION POINT WAS NORTH AND TOO HIGH.
LAST COORDINATES = 6,1,1

TRIAL #4
EAST/WEST: 6
NORTH/SOUTH: 0
DEPTH: 2

B O O M ! ! YOU FOUND IT IN 4 TRIES!
ANOTHER GAME? (NO)

PROGRAM LISTING

```

1000.
1010.
1020.
1030. *SUB - SEARCH*
1040. BY MARC GLADSTEIN
1050 NT = 0
1060 CLEAR
1070 BC = 0
1080 FC = -65
1090 PRINT
1100 PRINT "AREA DIMENSIONS:"
1110 INPUT "EAST/WEST: " G
1120 INPUT "NORTH/SOUTH: " H
1130 INPUT "DEPTH: " I
1140 IF G < 1 G = 1
1150 IF H < 1 H = 1
1160 IF I < 1 I = 1
1170 INPUT "NUMBER OF SHOTS: " N
1180 IF N < 1 N = 1
1190 PRINT
1200 PRINT
1210 PRINT "DO YOU WANT INSTRUCTIONS?"
1220 K = KP
1230 IF K = "N" GOTO 1440
1240 CLEAR
1250 PRINT "          *SUB SEARCH*"
1260 PRINT "YOU ARE CAPTAIN OF THE DESTROYER USS BINARY #11010."
1270 PRINT "AN UNFRIENDLY RUSSIAN SUB HAS BEEN BUGGING YOU,"
1280 PRINT "YOUR MISSION IS TO SEEK AND DESTROY IT."
1290 PRINT "YOU HAVE ", #1, N, "SHOTS,"
1300 IF N = 1 PRINT "."
1310 IF N > 1 PRINT "S."
1320 PRINT
1330 PRINT "          PRESS GO"
1340 IF KP = 13 CLEAR
1350 PRINT
1360 PRINT "          SUB SEARCH CONT."
1370 PRINT "INDICATE EXPLOSION POINT WITH X,Y,Z COORDINATES,"
1380 PRINT "THE FIRST TWO NUMBERS ARE THE SURFACE COORDINATES,"
1390 PRINT "THE THIRD IS FOR THE DEPTH .(FIRST TWO ARE REGULAR X,Y"
1400 PRINT "COORDINATES,THE THIRD IS THE OTHER DIMENSION)."
1410 PRINT "MAY THE FORCE BE WITH YOU!"
1420 PRINT "          PRESS GO TO START"
1430 IF KP#13 GOTO 1430
1440 CLEAR
1450 A = RND (G)-1
1460 B = RND (H)-1
1470 C = RND (I)-1

```

CONTINUED ON NEXT PAGE

PROGRAM LISTING
(CONTINUED)

```
1480 FOR D=1 TO N
1490 PRINT
1500 PRINT "TRIAL #",#1,D
1510 INPUT "EAST/WEST:"X
1520 INPUT "NORTH/SOUTH:"Y
1530 INPUT "DEPTH:"Z
1540 IF ABS (X-A)+ABS (Y-B)+ABS (Z-C) = 0 GOTO 1650
1550 CLEAR
1560 GOSUB 1740
1570 PRINT #1,"LAST COORDINATES ",X,"",Y,"",Z
1580 PRINT
1590 NEXT D
1600 PRINT
1610 PRINT "YOU'VE BEEN TORPEDOED! ABANDON SHIP!"
1620 PRINT #1,"THE SUBMARINE WAS AT ",A,"",B,"",C
1630 GOTO 1680
1640 PRINT
1650 PRINT "B O O M ! ! YOU FOUND IT IN ",#1,D," TRIE",
1660 IF D=1 PRINT "I"
1670 IF D>1 PRINT "S!"
1680 PRINT
1690 PRINT
1700 PRINT "ANOTHER GAME?"
1710 K=KP
1720 IF K#89 GOTO 1850
1730 GOTO 1170
1740 PRINT "SONAR SENSES EXPLOSION POINT WAS ",
1750 IF Y > B PRINT "NORTH",
1760 IF Y < B PRINT "SOUTH",
1770 IF X > A PRINT "EAST",
1780 IF X < A PRINT "WEST",
1790 IF (Y#B)+(X#A) PRINT " AND ",
1800 IF Z > C PRINT "TOO LOW."
1810 IF Z < C PRINT "TOO HIGH."
1820 IF Z=C PRINT "DEPTH A - OK."
1830 PRINT
1840 RETURN
1850 PRINT "OK - HOPE YOU ENJOYED YOURSELF."
1860 NT=3
1870 BC=7
1880 FC=0
1890 CLEAR
1900 PRINT
1910 PRINT "BALLY BASIC"
```

END OF PROGRAM (E.O.P.).

NOTES

1. THE SAMPLE RUN IS NOT EXACTLY THE SAME AS THE REAL PROGRAM RUN THAT I T COPIED IT FROM , OF COURSE , NO VARIABLES WERE CHANGED, BUT THE BIGGEST DIFFERENCE (AND ALSO JUST ABOUT THE ONLY ONE) WAS THAT I DID NOT DO THE LINE SPACING PROPERLY (E.G. WHERE THE PROGRAM MIGHT HAVE DONE A 'PRINT;PRINT' , I MIGHT HAVE DONE ONLY ONE 'PRINT').
2. ON THE FLOWCHART I DID NOT PUT IN ALL OF THE THINGS THAT THE PROGRAM DOES , ONLY THE MAJOR THINGS (E.G. PRINTING OUT THE INSTRUCTIONS , THE SONAR SUBROUTINE , ETC.).
3. A SUGGESTION FOR MODIFYING THE PROGRAM:
MAKE THE SUB NONSTATIONARY (IT WOULD MOVE) , HAVE IT MOVE ONESSPACE IN ANY DIRECTION OR NOT MOVE AT ALL (THIS WOULD MAKE IT MORE LIKE ERIBB TRIBBLE , A COMPUTER GAME).
4. SOON I WILL BE BRINGING OUT MORE PROGRAMS , MY NEXT ONE BEING A PROGRAM TO CALCULATE SINE , COSINE , TAGENT , SQUARE ROOT , SQUARE , POWERS AND I'M ADDING DIVISION WITH A QUOTIENT IN FRACTIONS. PRICE - 95¢ , PROGRAM HAS SAME QUALITY DOCUMENTATION (PAPERS) AS SUB SEARCH EXEPT NO FLOWCHART. DON'T ORDER NOW , WAIT TILL IT COMES OUT IN THE ARCADIAN.

MAP OF 10x10x10 GRID AREA

THIS IS A MAP OF AN AREA OF SEAWATER THAT YOU WILL BE SEARCHING IN , YOU MAY SELECT A CUBE LARGER OR SMALLER THAN THIS ONE , TO DO THIS TYPE IN THE APPROPRIATE NUMBERS FOR THE AREA DIMENSIONS (SEE PAGE #1). WHEN YOU ARE USING THIS MAP NOTE THAT IF YOUR SHOT HIT EAST OF THE SUB , YOUR NEXT SHOT MUST BE FARTHER WEST. THE SAME GOES FOR NORTH/SOUTH COORDINATES , IN OTHER WORDS IF YOUR SHOT HIT WEST OF THE SUB , THEN YOUR NEXT SHOT MUST BE FARTHER EAST. THIS PRINCIPLE ALSO GOES FOR THE DEPTH. IF YOU ARE CONFUSED , SEE SAMPLE RUN ON PAGE #3.

